

Lab::Measurement – measurement control with Perl

S. Reinhardt¹, C. E. Lane², C. Butschkow¹, A. Iankilevitch¹, A. Dirnaichner¹, and A. K. Hüttel¹

¹*Institute for Experimental and Applied Physics, University of Regensburg, 93040 Regensburg, Germany*

²*Department of Physics, Drexel University, 3141 Chestnut Street, Philadelphia, PA 19104, USA*

Flexible measurement needed?!

- Tired of following your wires in square meters of LabVIEW diagrams?
- Tired of clumsy string handling and low-level driver functions in your loong C program?
- Use a text processing language to manage your measurement! Use Perl!

```
#!/usr/bin/env perl
## Read out SR830 Lock In Amplifier at GPIB address 13
use 5.010;
use Lab::Moose;

my $lia = instrument(
    type           => 'SR830',
    connection_type => 'LinuxGPIB',
    connection_options => { pad => 13 }
);

my $amp = $lia->get_amplitude();
say "Reference output amplitude: $amp V";

my $freq = $lia->get_freq();
say "Reference frequency: $freq Hz";

my $r_phi = $lia->get_rphi();
my ( $r, $phi ) = ( $r_phi->{r}, $r_phi->{phi} );
say "Signal: r=$r V, phi=$phi degree";
```

Currently supported hardware



Hardware driver backends:


- NI-VISA (MS Windows) and all hardware supported by it
- LinuxGPIB and all hardware supported by it
- TCP connection, generic network socket (both Linux and MS Windows)
- USB-TMC lightweight driver (Linux, libusb)
- VXI-11 lightweight driver (Linux, libtirpc)
- Zurich Instruments LabOne API (both Linux and MS Windows)

Growing number of high-level drivers (more are very easy to add):

- Multimeters: HP / Agilent / Keysight
- DC sources: Yokogawa / Keithley / Keysight
- Lock-in amplifiers: Stanford Research / Signal Recovery / Zurich Instruments
- Temperature controllers: Lakeshore / Oxford Instruments
- RF / microwave sources, spectrum analyzers, VNAs: Rohde & Schwarz, HP / Agilent, Rigol
- and many more...

Key facts

- Open source / free software
 - <https://www.labmeasurement.de/>
 - License: same as Perl (GPL-1+ or Artistic)
 - Releases on CPAN, development on Github
 - Contributors and cooperations welcome!
- 



Real world measurement

- Ferromagnetic resonance measurement with vector network analyzer (VNA)
- Outer loop: continuous sweep of magnetic field with OI Mercury iPS magnet controller
- Inner loop: VNA transmission measurement at multiple discrete microwave frequencies

```
#!/usr/bin/env perl
use 5.010;
use Lab::Moose;

# Define the two instruments
my $sps = instrument(
    type          => 'OI_Mercury::Magnet',
    connection_type => 'Socket',
    connection_options => { host => '192.168.3.15' },
);

my $vna = instrument(
    type          => 'RS_ZVA',
    connection_type => 'VXI11',
    connection_options => { host => '192.168.3.27' },
);

# Set VNA's IF filter bandwidth (Hz)
$vna->sense_bandwidth_resolution( value => 1 );

# The outer sweep: continuous magnetic field sweep
my $field_sweep = sweep(
    type          => 'Continuous:Magnet',
    instrument     => $sps,
    from          => 2,      # Tesla
    to            => 0,      # Tesla
    rate          => 0.01,   # Tesla/min
    start_rate    => 1,      # Tesla/min (rate to approach start point)
    interval      => 0,      # run slave sweep as often as possible
);

# The inner sweep: set frequency to 1GHz, 2GHz, ..., 10GHz
my $fqr_sweep = sweep(
    type          => 'Step:Frequency',
    instrument     => $vna,
    from          => 1e9,
    to            => 10e9,
    step          => 1e9
);

# The data file: gnuplot-style, VNA data prefixed with B
my $datafile = sweep_datafile(
    columns => [ 'B', 'f', 'Re', 'Im', 'r', 'phi' ] );

# Add a live plot of the transmission amplitude
$datafile->add_plot( x => 'B', y => 'r' );

# Define the measurement instructions per (B,f) point
my $meas = sub {
    my $sweep = shift;

    say "frequency f: ", $sweep->get_value();
    my $field = $sps->get_field();

    # this is not really a VNA "sweep", but only a
    # point measurement at one frequency
    my $pdl = $vna->sparam_n_sweep( timeout => 10 );

    # Record the result, prefixed with B
    $sweep->log_block(
        prefix => { field => $field },
        block   => $pdl
    );
};

# And go!
$field_sweep->start(
    slave      => $fqr_sweep,
    datafile   => $datafile,
    measurement => $meas,
    folder     => 'Magnetic_Resonance',
);
```

Output files


```
simon@dilfridge ~/FMR$ ls Magnetic_Resonance_001/
META.yml  data.dat  data.png  magnet-resonance.pl
```

- *META.yml*: various metadata (host, user, date, L::M version, command line arguments)
- *data.dat*: measured data, in tab-separated Gnu-plot format
- *data.png*: live plot at the end of the measurement, as a png image
- *magnetic-resonance.pl*: archival copy of the measurement script

Advanced sweep features

- Multidimensional sweeps, e.g. 3D sweep: creating one 2D datafile for each step of the outermost sweep
- Log arrays and matrices of data (PDLs). Useful for spectrum analyzers, VNAs, oscilloscopes
- Extensive support for live plots via gnuplot: line plots (2D data) and color maps (3D data)
- Customizing live plots: access to all gnuplot plot and curve options via `PDL::Graphics::Gnuplot`


Layer structure



- Modular structure. Easy to extend with new instrument drivers and connection types
- Abstract IO layer, makes instrument drivers independent of hardware backends

High-level sweep framework

- Modern Perl implementation; use state of the art object-oriented programming



- Separate classes for sweeps, datafiles, datafolders, and plots
- Most operational details of the sweeps are implemented in subclasses of the Lab::Moose::Sweep base class
- High modularity: *very* easy to extend with new functionality

Recent improvements

- Extensive tutorial
- New drivers
 - Oxford Instruments Mercury magnet controller and level meter
 - Oxford Instruments Triton system control
 - Lakeshore 340 temperature controller
 - Zurich Instruments MFLI lock-in amplifier and MFIA impedance analyzer
 - Rigol DSA815, HP 8596E Spectrum Analyzer
 - Keithley 2400, Keysight B2901A “sourcemeter”
 - HP 34420A nanovoltmeter
- More configurable live plots
- Compatibility with newer versions of Zurich Instruments LabOne API

Reference / Cite as

“Lab::Measurement — a portable and extensible framework for controlling lab equipment and conducting measurements”, S. Reinhardt *et al.*, Comp. Phys. Comm. **234, 216 (2019)**